LA-6076-MS
Informal Report
c.3

UC-2
Reporting Date: August 1975
Issued: September 1975

Lognormal Damage Function with Offset Circular Normal Target and Weapon Locations: Mean and Other Moments

by

L. A. Bruckner

los alamos
scientific laboratory
of the University of California
LOS ALAMOS, NEW MEXICO 87545

An Affirmative Action/Equal Opportunity Employer

UNITED STATES
ENERGY RESEARCH AND DEVELOPMENT ADMINISTRATION
CONTRACT W-7405-ENG. 36

In the interest of prompt distribution, this report was not edited by the Technical Information staff.

Printed in the United States of America. Available from
National Technical Information Service
U S Department of Commerce
5285 Port Royal Road
Springfield, VA 22151
Price: Printed Copy $4.00 Microfiche $2.25

LOGNORMAL DAMAGE FUNCTION WITH OFFSET CIRCULAR NORMAL TARGET AND WEAPON LOCATIONS: MEAN AND OTHER MOMENTS

by

L. A. Bruckner

ABSTRACT

If target location and weapon impact point have independent circular normal probability distributions, the probability density function of the distance between target and weapon impact can be determined. If a damage function is introduced as a measure of the probability of damage to a target, the expected probability of damage can then be computed. This report shows how to compute and use mean and higher moments of the probability of damage for a lognormal damage function. Moments for other damage functions can be similarly computed.

I. INTRODUCTION

If the distance, R, between a target and ground zero (GZ) of a weapon is subject to random uncertainties as to location of both the target and GZ, the probability density function, $f_R(r)$, of R can sometimes be found explicitly. In these cases the probability that the weapon impacts within a distance d of the target is given by

$$P(R \leq d) = \int_0^d f_R(r)\, dr.$$

Commonly, damage to a target at a distance r from GZ is also subject to uncertainties, that is, the destruction of a target at distance r from GZ is measured by a damage function G(r) which gives the probability of a target at this distance being damaged. Then the uncertainties due to location and damage can be combined as follows:

$$\text{Expected } P(\text{damage}) = \int_0^\infty f_R(r)\, G(r)\, dr.$$

This report will consider the case in which both the target and weapon ground zero have circular normal distributions about possible different locations and the damage function is the lognormal damage function. The computer routine developed allows easy substitution of other damage functions.

II. CIRCULAR NORMAL CASE

If the location of a target is not known exactly it is sometimes assumed that its true location has a circular normal probability distribution about some point in a plane. Letting that point be the origin of a superimposed coordinate system, then

$$P_T(x,y) = \frac{1}{2\pi\, \sigma_T{}^2} \exp\left(-\frac{x^2+y^2}{2\sigma_T{}^2}\right)$$

gives the probability density function of the target's location. σ_T is the standard deviation in both the x and y directions. The probability that the target lies in some region A of the plane is then given by

$$P(\text{target lies in A}) = \iint_A p_T(x,y)\,dx\,dy.$$

Then, if the weapon ground zero (GZ) has a circular normal distribution about a desired ground zero (DGZ), say (a,b), with standard deviation σ_w, the probability density of GZ is

$$p_w(x,y) = \frac{1}{2\pi\,\sigma_w^2}\exp\left(-\frac{1}{2\sigma_w^2}\left[(x-a)^2+(y-b)^2\right]\right).$$

The distance, R, between the actual target location and GZ is a random variable, $R = \sqrt{(x_w-x_T)^2 + (y_w-y_T)^2}$. It can be shown that the probability density function of R has the form

$$f_R(r) = \frac{2r}{\sigma^2}\, f_{R^2/\sigma^2}\left(\frac{r^2}{\sigma^2}\right) \qquad \text{for } r > 0$$

where $\sigma^2 = \sigma_T^2 + \sigma_w^2$ and $f_{R^2\sigma^2}$ is the density of a noncentral χ^2 distribution with two degrees of freedom and noncentrality parameter $\tau^2 = (a^2+b^2)/\sigma^2$. If the weapon is aimed at the origin (the mean target location) of our system, the noncentrality parameter disappears (Appendix A) and $f_R(r)$ is simply a function of the χ^2 distribution.

III. THE LOGNORMAL DAMAGE FUNCTION

The damage to a target from a particular weapon depends not only on R, the distance between the target and the GZ, and the weapon radius, WR, but also on a random factor. This random factor will be incorporated by using a lognormal damage function. The probability of damage to a target at distance r is given by

$$G(r) = \int_{-\infty}^{z(r)} n(y)\,dy$$

where

n(y) is the standardized normal density

$$z(r) = \frac{1}{\beta}\,\ell n\left(\frac{WR\ e^{-\beta^2}}{r}\right) = -\beta + \frac{1}{\beta}\,\ell n\left(\frac{WR}{r}\right)$$

$$\beta^2 = -\,\ell n\,(1 - \sigma_d^2)$$

$WR\cdot\sigma_d$ = the standard deviation of the lognormal distribution.

A more complete discussion of the lognormal damage function is contained in Appendix B.

IV. THE PROBABILITY OF DAMAGE

Combining the results of the previous sections, the expected (mean) probability of damage, M, is given by

$$M = \int_0^\infty f_R(r)\,G(r)\,dr$$

$$= e^{-\tau^2/2}\int_0^\infty e^{-\mu}\, I_o\,(\tau\sqrt{2\mu})\; G(\sqrt{2\mu}\sigma)\,d\mu$$

where

$$I_o\,(\tau\sqrt{2\mu}) = \sum_{j=o}^{\infty}\frac{\left(\frac{1}{2}\tau^2\ \mu\right)^j}{(j!)^2}$$

is a modified Bessel function of order zero and $\mu = 0.5(r/\sigma)^2$.

To find the standard deviation of the probability of damage, compute the variance using the formula $VAR = E(G(R)^2)-M^2$. The term $E(G(R)^2)$ is found by replacing G(r) with $G(r)^2$ in the preceding formulas. Then $STD = VAR^{1/2}$. A discussion of the use of the STD and other moments is contained in Appendix C.

It is easy to show that the value of M and STD calculated from a particular set of values of σ, WR, and DIST (= a^2+b^2) remains unchanged if these values are replaced by K·σ, K·WR, and K·DIST where K is any positive constant.

V. RESULTS

The expected probability of damage, M, and the standard deviation of the damage, STD, were computed using a computer program run on a Control Data Corporation 6600 under the KRONOS time-sharing system. A program listing with notes is contained in Appendix D. Figure 1 presents graphs of M vs DIST/WR for values of SIG/WR equal to 0, 0.5, 1, and 2 where DIST is the distance between the mean target location and the weapon aim point, WR is the weapon radius, and SIG is the combined target-weapon standard deviation, σ.

To use these graphs, compute SIG/WR and DIST/WR and read (interpolate) M from the appropriate curve(s). Table I contains some values of M = MEAN and STD along with the coefficient of variation, COEF=STD/MEAN, for WR = 1000.



Figure 1. For values of offset distance/WR less than 1, the expected probability of damage decreases as SIGD increases.

TABLE I

MOMENTS OF THE PROBABILITY OF DAMAGE TO A TARGET

SIG	DIST	SIGD	MEAN	E(G**2)	VAR	STD	COEF
500.0	0.0	.3	.777	.682	.078	.280	.360
500.0	0.0	.7	.470	.278	.057	.239	.508
500.0	500.0	.3	.651	.537	.113	.336	.517
500.0	500.0	.7	.384	.203	.055	.235	.611
500.0	1000.0	.3	.369	.251	.115	.340	.922
500.0	1000.0	.7	.225	.083	.032	.179	.794
500.0	2000.0	.3	.030	.009	.008	.088	2.904
500.0	2000.0	.7	.057	.005	.002	.046	.815
1000.0	0.0	.3	.365	.272	.139	.373	1.024
1000.0	0.0	.7	.230	.099	.046	.215	.931
1000.0	500.0	.3	.333	.246	.135	.368	1.103
1000.0	500.0	.7	.214	.089	.044	.209	.979
1000.0	1000.0	.3	.254	.181	.117	.342	1.346
1000.0	1000.0	.7	.171	.066	.036	.191	1.115
1000.0	2000.0	.3	.085	.053	.046	.214	2.517
1000.0	2000.0	.7	.076	.019	.014	.117	1.549
2000.0	0.0	.3	.115	.079	.066	.256	2.238
2000.0	0.0	.7	.087	.029	.021	.145	1.661
2000.0	500.0	.3	.111	.076	.064	.253	2.275
2000.0	500.0	.7	.085	.028	.021	.144	1.681
2000.0	1000.0	.3	.102	.070	.060	.244	2.387
2000.0	1000.0	.7	.079	.026	.019	.139	1.748
2000.0	2000.0	.3	.073	.049	.044	.209	2.886
2000.0	2000.0	.7	.060	.018	.014	.120	2.021

WR = 1000.

APPENDIX A

Let $R = \sqrt{(X_W - X_T)^2 + (Y_W - Y_T)^2}$ where (X_W, Y_W) has a circular normal probability distribution about the point (a,b) and standard deviation σ_W, and (X_T, Y_T) has a circular normal probability distribution about the origin with standard deviation σ_T. Then the probability density function of R, $f_R(r)$, can be shown to be

$$f_R(r) = \frac{2r}{\sigma^2} \left\{ \frac{e^{-\frac{1}{2}\left(\tau^2 + \frac{r^2}{\sigma^2}\right)}}{2} \sum_{j=0}^{\infty} \left(\frac{\tau r}{2\sigma}\right)^{2j} \frac{1}{(j!)^2} \right\} \qquad r > o$$

where $\sigma^2 = \sigma_T^2 + \sigma_W^2$. The term in the brackets is the probability density function of a noncentral χ^2 distribution with noncentrality parameter $\tau^2 = (a^2+b^2)/\sigma^2$ evaluated at r^2/σ^2. If (a,b) = (0,0) (i.e., there is no offset) then $\tau = 0$ and

$$f_R(r) = \frac{2r}{\sigma^2} \cdot f_{\chi_2^2}\left(\frac{r^2}{\sigma^2}\right) \qquad r > 0$$

where ${\chi_2}^2$ is a chi-square distribution with two degrees of freedom.

Since R^2/σ^2 has the above noncentral χ^2 distribution, the mean and variance of R^2 are $\sigma^2(2+\tau^2)$ and $4\sigma^4(1+\tau^2)$ respectively. Letting $R = K(R^2) = \sqrt{R^2}$, it can be shown from the Taylor series expansion of $K(R^2)$ that

$$E(R) \approx K(\mu_1) + \frac{K''(\mu_1)}{2} \sigma_1^2$$

$$V(R) \approx (K'(\mu_1))^2 {\sigma_1}^2$$

where μ_1, σ_1 are the mean and standard deviation of R^2 respectively. Letting $D^2 = a^2+b^2$,

$$E(R) \approx \sqrt{2\sigma^2+D^2} - .5(\sigma^4+\sigma^2\sigma D^2)/(2\sigma^2+D^2)^{3/2}$$

$$V(R) \approx (\sigma^4+\sigma^2D^2)/(2\sigma^2+D^2) .$$

If there is no offset, $E(R) \approx 1.237\sigma$ and $V(R) \approx .5\sigma^2$.

APPENDIX B

A random variable Y has a lognormal distribution if, and only if, the random variable $X = \ln Y$ is normally distributed. Then, if a,b are the mean and standard deviation respectively of X

$$f_y(y) = \frac{1}{y} \frac{1}{\sqrt{2\pi}\, b} \exp\left[-\frac{(\ln y - a)^2}{2b^2}\right], \quad y > 0 .$$

$$\mu_y = \exp\,[a+b^2/2]$$

$${\sigma_y}^2 = (\exp\,[b^2+2a])(\exp\,[b^2]-1) .$$

It can be shown that the lognormal damage function has the form

$$G(r) = \frac{1}{\sqrt{2\pi}\,\beta} \int_r^{\infty} \frac{1}{y} \exp\left[-\frac{(\ln y - (\ln WR - \beta^2))^2}{2\beta^2}\right] dy$$

where $\beta^2 = -\ln(1-\sigma_d^2)$.

Hence, the mean and standard deviation of this lognormal distribution are

$$\text{MEAN} = WR\sqrt{(1-\sigma_d^2}$$

$$\text{STD} = (WR)\cdot\sigma_d .$$

Note that as σ_d approaches zero, the damage function approaches the "cookie cutter" damage function of radius WR.

APPENDIX C

The variance of a random variable R having mean, $E_R(R) = m$, is $VAR = E_R(R-m)^2$. This can be shown to be equivalent to $VAR = E_R(R^2)-m^2$. If $Z = G(R)$ is a function of R, then $VAR(Z) = E_R(Z^2)-E_R(Z)^2$. If the mean and variance of Z exist, Chebyshev's Inequality holds:

$$P(|Z-E(Z)|<\varepsilon) > 1-VAR(Z)/\varepsilon^2 . \qquad (C-1)$$

If Z assumes values in the interval [0,1] only, it can be shown (M. Loeve, Probability Theory (D. Van Nostrand, 1963), p. 158) that for s > 0,

$$E(Z^s)-a^s < P(Z > a) < E(Z^s)/a^s . \qquad (C-2)$$

When Z=G(R) is a damage function the above inequalities hold. The higher order moments may be computed using the included program.

Example: If WR=1000, SIG=5000, SIGD=0.3 and D=0, E(Z)=0.777, EZ^2=0.682, VAR(Z)=0.078 and EZ^5=0.561. Inequality (C-1) gives $P(G(R) \geq 0.25) \geq 0.72$ while (C-2) does not do as well (for integer s). However, when computing $P(G(R) \geq 0.5)$, inequality (C-1) is useless while (C-2) for s=5 yields $P(G(R) \geq 0.5) \geq 0.53$. Hence, if 100 targets were independently attacked by 100 weapons with the given characteristics the probability of damaging at least 50 targets would be at least 0.53.

APPENDIX D

```
      SUBROUTINE MOMENTS(WR,SIGD,SIG,DIST,ITH,P)
C     WR = WEAPON RADIUS.
C     SIGD = STANDARD DEVIATION OF THE LOGNORMAL DISTRIBUTION/WR.
C     SIG = COMBINED STANDARD DEVIATION OF THE TARGET
C           LOCATION AND THE WEAPON DELIVERY.
C     DIST = THE DISTANCE BETWEEN THE MEAN TARGET LOCATION
C            AND THE WEAPON AIM POINT.
C     P = THE EXPECTED PROBABILITY OF DAMAGING THE TARGET.
C     ITH = THE MOMENT OF THE DAMAGE FUNCTION,G(R),
C          TO BE CALCULATED. IF THE VARIANCE IS DESIRED
C          CALL MOMENTS WITH IPOWER = 1 AND 2. THEN USE
C          THE FORMULA   VARIANCE = SECOND MOMENT MINUS
C          THE FIRST MOMENT SQUARED.
C
C     THIS ROUTINE COMPUTES THE I TH MOMENT OF
C     THE  PROBABILITY  OF  DAMAGE
C     TO A TARGET FROM A WEAPON WHERE THE DAMAGE
C     FUNCTION IS G(R). THE TARGET LOCATION
C     AND DELIVERY ERROR ARE ASSUMED TO HAVE INDEPENDENT
C     CIRCULAR NORMAL PROBABILITY DISTRIBUTIONS.
C
C     TO CHANGE THE DAMAGE FUNCTION SIMPLY CHANGE THE
C     APPROPIATE CARDS (INDICATED BY A * IN COLUMN 75)
C     IN SUBROUTINES FUNCTION F  AND DECIDE.
```

```
      COMMON /ADAM1/ S,SD,RAD,D,IPOWER
      EXTERNAL F
      DIMENSION WK(121),IW(5)
      S=SIG
      SD=SIGD
      RAD=WR
      D=DIST
      IPOWER=ITH
      CALL DECIDE(S,SD,RAD,D,IPOWER,INDEX,XLO,XHI,P)
C
C     XLO = THE LOWER LIMIT OF INTEGRATION
C     FOR CONSERVATISM,SET XLO=0.
C     XHI = THE UPPER LIMIT OF INTEGRATION
C     INDEX CONTROLS WHETHER INTEGRATION IS NESESSARY OR NOT.
C
      IF(INDEX.EQ.1) RETURN

      P=0.
      XLO=0.
C
C     THE FOLLOWING, THROUGH STATEMENT 10, DOES THE
C     ACTUAL INTEGRATION. RE AND AE CONTROL THE
C     ACCURACY. THE INTERVAL OF INTEGRATION IS
C     DIVIDED INTO 10 SUBINTERVALS. HENCE EACH IS
C     ABOUT A STANDARD DEVIATION IN LENGTH.
C
      T=XLO
      RE=1.E-04
      AE=RE
      IFLAG=1
      NEQ=1
      TOUT=XLO
      NINTV=10
      TINCR=(XHI-XLO)/NINTV
      DO 10 I=1,NINTV
      TOUT=TOUT+TINCR
      CALL ODE(F,NEQ,P,T,TOUT,RE,AE,IFLAG,WK,IW)
C     OUTPUT,TOUT,P,IFLAG
   10 CONTINUE

      RETURN
      END


      SUBROUTINE DECIDE(S,SD,RAD,D,IPOWER,INDEX,ULO,UHI,P)
C
C     THIS ROUTINE DECIDES  WHETHER INTEGRATION IS NECESSARY
C     OR NOT. IF SO IT DETERMINES THE LIMITS OF INTEGRATION,ULO
C     AND UHI.IT FIRST COMPUTES THE MEAN AND STANDARD DEVIATION
C     OF THE VARIABLE R AND THEN COMPUTES THE MEAN PLUS AND
C     MINUS 5 STANDARD DEVIATIONS. ULO AND UHI REPRESENT
C     THE FINAL LIMITS OF INTEGRATION.
C     INTEGRATION IS REQUIRED IF INDEX = 0.
C
C     D IS THE DESIRED OFFSET
C     IF THE OFFSET IS ZERO AND SIG IS ZERO, P=1.
C     IF THE OFFSET IS ZERO AND SIG > 0.,INTEGRATE.
C     IF THE OFFSET>0,AND S IS SMALL IN COMPARISON TO D,
C        TREAT THE PROBLEM DETERMINISTICALLY. SET P=G(D).
C
      ZZ(R)=B1*(-B2+ALOG(RAD/R))                                      *
      AB(R)=ABS(ZZ(R))                                                *
      G(R)=(.5*(1.+ZZ(R)/AB(R) * ERF(AB(R)/SQRT(2.))))**IPOWER        *
      B2=-ALOG(1.-SD**2)                                              *
      B1=1./SQRT(B2)                                                  *
C
      INDEX=1
      P=1.
      IF(D.EQ.0.) GO TO 30
      IF(S/D.LE..01) GO TO 20
```

```
C
C     COMPUTE RMEAN, SIGR, MEAN PLUS/MINUS 5*SIGR
C
      F1=SQRT(2.*S*S+D*D)
      F2=S**4+D*D*S*S
      RMEAN=F1-.5*F2/(F1**3)
      SIGR=SQRT(F2)/F1
      RLO=AMAX1(0.,RMEAN-5.*SIGR)
      RHI=RMEAN+5.*SIGR
C
C     CHECK TO SEE IF G IS ESSENTIALLY CONSTANT OVER THE
C     RANGE RLO TO RHI. IF IT IS DO NOT INTEGRATE AND SET
C     P=G(RMEAN). IF INTEGRATION IS NESESSARY, MODIFY
C     LIMITS OF INTEGRATION TO CORRESPOND TO TRANSFORMATIONS
C     MADE IN THE INTEGRAL - U=(1/2)*(R/S)**2.
C
      GHI=G(RHI)
      IF(RLO.EQ.0.) GO TO 10
      GLO=G(RLO)
      GDIF=GLO-GHI
      GO TO 15
   10 GDIF=1.-GHI
   15 IF(GDIF.GE.1.E-04) INDEX=0
      P=G(RMEAN)
      ULO=.5*(RLO/S)**2
      UHI=.5*(RHI/S)**2
      RETURN
   30 IF(S.GT.0.) INDEX = 0
C     IF D=0., RMEAN=1.24*S AND SIGR=.71*S
C     HENCE, UHI=RMEAN+5*SIGR=5.*S(APPROX)
      ULO=0.
      UHI=5.*S
      RETURN
   20 P=G(D)
      RETURN
      END

      SUBROUTINE F(U,Y,YP)
C
C     THIS ROUTINE COMPUTES THE VALUE OF THE INTEGRAND.
C     U IS THE VARIABLE OF INTEGRATION
C
      COMMON /ADAM1/ S,SD,RAD,D,IPOWER
      ZZ(R)=B1*(-B2+ALOG(RAD/R))                                    *
      AB(R)=ABS(ZZ(R))                                              *
      G(R)=(.5*(1.+ZZ(R)/AB(R) * ERF(AB(R)/SQRT(2.))))**IPOWER      *
      FF(X)=H(TAU2,X)*G(SQRT(2.*X)*S)
      B2= -ALOG(1.-SD**2)                                           *
      B1=1./SQRT(B2)                                                *
      TAU2=(D/S)**2
      YP=1.
      IF(U.LE.0.)RETURN
C
C     IF THE OFFSET DISTANCE, D, IS ZERO, THE INTEGRAND SIMPLIFIES.
C
      IF(D.EQ.0.) GO TO 9
      YP = FF(U)
      RETURN
    9 YP=G(SQRT(2.*U)*S)*EXP(-U)
      RETURN
      END
```

```
      FUNCTION H(TAU2,U)
C
C     THIS ROUTINES COMPUTES THE VALUE OF THE PROBABILITY
C     DENSITY FUNCTION OF R. IT FIRST COMPUTES
C     THE VALUE OF A MODIFIED BESSEL FUNCTION OF ORDER ZERO
C     USING A POLYNOMIAL APPROXIMATION. ERROR IS
C     LESS THAN 1.E-07. THE RESULT IS THEN MULTIPLIED BY
C     EXP(-TAU2/2 - U) TO FORM THE DENSITY.
C
      X=SQRT(2.*U*TAU2)
      T=X/3.75
      IF(X.GT.3.75) GO TO 10
      S=T**2
      BESL1=1.+S*(3.5156229+S*(3.0899424+S*(1.2067492+S*(.2659732+
     1 S*(.0360768+S*.0045813)))))
      H=EXP(-TAU2/2.-U)*BESL1
      RETURN
   10 S=1/T
      BESL1=.39894228+S*(.01328592+S*(.00225319+S*(-.00157565+
     1 S*(.00916281+S*(-.02057706+S*(.02635537+S*(-.01647633+
     2 S*.00392377)))))))
      H=EXP(-TAU2/2.-U+X)/SQRT(X) * BESL1
      RETURN
      END
```